
1 ハンズオンの概要
ProVerifは暗号プロトコルの形式検証ツールです．暗号プロトコルと検証したい安全性要件を入力する
と，そのプロトコルが（検証可能な範囲内で）要件を満たしているか，満たしていないかを出力してく
れます．TLS 1.3や 5G認証プロトコルなどの様々なプロトコルの検証に用いられており，その安全性の
向上に貢献しています 1．
このハンズオンでは，ProVerifの簡単な使い方を習得することをゴールとします．具体的には，ProVerif

における
• 暗号プロトコルの記述方法
• 安全性要件の記述方法
• 検証結果の読み方

の基礎を理解してもらうことをゴールとします．時間の都合上，細かい文法や記述方法については扱い
ません．興味を持っていただけた方は ProVerifの公式マニュアル [BSCS21]などを参照してください．
なお，本テキストで扱うコードは FWSのWebページ
• https://www.iwsec.org/fws/2023/

から入手可能です．

∗中林美郷（NTT），花谷嘉一（東芝），吉田真紀（NICT），米山一樹（茨城大学）
1ProVerif の公式ページ [Bla21] に関連論文の一覧があります．検証例についてはそちらをご覧ください．

2 準備
2.1 オンラインデモの使い方
このハンズオンでは ProVerifのオンラインデモ [MB21]を使用します．

• http://proverif20.paris.inria.fr/index.php

図 1: ProVerifオンラインデモの画面．

右側のフォームにコードを入力し，Verifyボタンを押すと検証が始まり，検証結果が出力されます．検
証結果の最下部に Verification summaryという部分があるので，そこを見ると各安全性要件（クエリ）
に対する検証結果が分かります（図 2）．trueが攻撃発見なし，falseが攻撃発見を表します．

図 2: ProVerifの出力の例．

ProVerif HTML output では，より詳細な出力を見ることができます．false の場合は ProVerif

HTML outputの Trace graphから発見された攻撃のトレース（図 3）を見ることができます．

図 3: ProVerifが出力する攻撃トレースの出力の例（一部）．

2.2 事前知識
2.2.1 暗号プロトコルとその安全性
暗号プロトコルとは，通信の機密性や完全性を保証するための通信プロトコルで，公開鍵暗号や電子署
名，ハッシュ関数などの暗号プリミティブを用いて構成されます．
鍵交換プロトコル，認証プロトコル，電子投票プロトコルなどがその一例です．このハンズオンでは

主に鍵交換プロトコルと認証プロトコルにフォーカスします．
鍵交換プロトコルは通信相手と安全に共通の鍵（セッション鍵）を共有するプロトコルです．通信路

の盗聴や改ざんができる攻撃者（アクティブな攻撃者）に対するセッション鍵の機密性などが主要な安
全性要件です．認証プロトコルは電子署名や共通の秘密を用いて通信相手が正しいことを確認するプロ
トコルです．認証が受理された場合その相手は必ず正しい通信相手であること（ここでは認証要件と呼
ぶことにします）などが主要な安全性要件です．通信相手の認証を行い，かつ鍵交換を行うプロトコル
を認証付き鍵交換プロトコルと呼びます．より詳しい情報は [BMS19]などを参照してください．

2.2.2 Dolev-Yaoモデル
ProVerifによる検証 2では，Dolev-Yaoモデル [DY83]という攻撃者モデルを考えています．Dolev-Yao
モデルでは暗号プリミティブは理想的なものとして扱われます．すなわち，共通鍵暗号方式では暗号文
に対応する復号鍵をもっているときに限り平文を手に入れることができ，電子署名方式では署名鍵を持っ
ているときに限り正しく検証される署名を作ることができると仮定されます．メッセージや鍵などの情
報を文字列ではなく記号的に扱うため，シンボリックモデルとも呼ばれています．その記号的な操作を
用いて，中間者攻撃やリプレイ攻撃，メッセージの順番を入れ替える攻撃などを捉えることができます．

2（小噺その１）ちなみに，ProVerif はいつ頃から使われていたかご存じでしょうか．2003 年頃から使われていたため，なん
と今年で約 20 歳です．結構昔からありますね．

3 ProVerifを用いた暗号プロトコルの安全性検証
ProVerifの入力コードは，主に次の 4つの部分から構成されます．

• 宣言．通信路や変数の型，プロトコル内で使用する暗号プリミティブを定義する．
• クエリ．検証したい安全性要件を記述する．
• マクロ．各プロトコル参加者の動作をサブプロセスとして記述する．
• メインプロセス．プロトコル実行を定義する．

3.1 最初の例
例として，次の二者間鍵交換プロトコル P を考えてみましょう．パーティAがナンスNAを生成してB

Generate NA

Alice

NA

Bob

Accept
skA = hash(NA)

Accept
skB = hash(NA)

図 4: 二者間鍵交換プロトコル P．

に送り，そのハッシュ値 hash(NA)をセッション鍵として受理するプロトコルです．プロトコルに詳し
い方はちょっと違和感を感じるかもしれませんが，どうぞお付き合いください．このプロトコルのアク
ティブな攻撃者に対するセッション鍵の機密性を検証するコードは次のようになります．

Code 1: 二者間鍵交換プロトコル P

1 (* Setting of channels and types *)
2 free c:channel.
3

4 (* Hash function *)
5 fun hash(bitstring):bitstring.
6

7 (* Symmetric key encryption *)
8 fun senc(bitstring, bitstring):bitstring.
9 reduc forall message:bitstring, key:bitstring; sdec(senc(message, key), key) = message.

10

11 (* Queries *)
12 free message_for_verification_A:bitstring[private].
13 query attacker(message_for_verification_A).
14 free message_for_verification_B:bitstring[private].
15 query attacker(message_for_verification_B).
16

17 (* Behavior of Alice *)
18 let Alice() =
19 new Na:bitstring;
20 out(c, Na);
21 let sk_Alice = hash(Na) in
22 out(c, senc(message_for_verification_A, sk_Alice)).
23

24 (* Behavior of Bob *)
25 let Bob() =
26 in(c, Na_recv:bitstring);
27 let sk_Bob = hash(Na_recv) in
28 out(c, senc(message_for_verification_B, sk_Bob)).
29

30 (* Start process *)
31 process
32 ((!Alice()) | !Bob())

1行目から 9行目が宣言部，11行目から 15行目がクエリ部，17行目から 28行目がマクロ部，30行
目から 32行目がメインプロセス部です．それぞれについて簡単に説明しますが，まずは重要なのはクエ
リ部とマクロ部です．それ以外の部分はおまじないとして捉えてもらっても構いません．

宣言部（1行目から 9行目）
2行目は通信路の定義です．cという盗聴・改ざん可能な通信路を定義しています．

4行目から 9行目は暗号プリミティブの定義です．5行目でハッシュ関数を，8行目・9行目で共通鍵
暗号を定義しています 3．ProVerifでは，暗号プリミティブはこのように暗号化関数などの暗号部品を
指定するコンストラクタ（5行目や 8行目）と復号条件などを指定するデストラクタ（9行目）のペアに
よって定義されます．ハッシュ関数は復号条件を指定する必要がないため，コンストラクタのみで定義
されます．ここで定義される関数は攻撃者も自由に使うことができます．

クエリ部（11行目から 15行目）
12行目・13行目はAのセッション鍵 skAの機密性を検証するクエリです．ProVerifの仕様の都合上，仮
の検証用メッセージを定義し（12行目），検証したい秘密情報を共通鍵としてそのメッセージを暗号化
した暗号文を通信路に流し（22行目），攻撃者が検証用メッセージを計算できるかどうかを見る（13行
目）というちょっと回りくどい方法で機密性の検証を表現しています．14行目・15行目も同様に，Bの
セッション鍵 skB の機密性を検証するクエリです．

マクロ部（17行目から 28行目）
17行目から 22行目が Aの動作を表しています．ナンスNA を生成し（19行目），それを通信路に流し
ます（20行目）．そしてセッション鍵を計算します（21行目）．22行目は上で述べた通りです．24行目
から 28行目は B の動作を表しており，ナンス NA を受け取り（26行目），セッション鍵を計算します
（27行目）．

メインプロセス部 (30行目から 32行目)

30行目から 32行目で，マクロ部で定義したサブプロセスを動かします．このように書くことで，無限個
のセッションを並列に動作させることができます．

演習 3.1. コード 1をオンラインデモに入力し，出力結果を確認してみましょう．

次のような結果が出力されたと思います．
• Query not attacker(message for verification A[]) is false.

• Query not attacker(message for verification B[]) is false.

両方とも falseが出力されているため，Aのセッション鍵の機密性，Bのセッション鍵の機密性それぞれ
に対する攻撃が発見されています．ついでに ProVerif HTML outputから攻撃トレースも見てみましょ
う．図 5と同じものが確認できると思います． トレースを見てみると，攻撃者は通信路を盗聴して NA

を盗み，セッション鍵 hash(NA)を計算しています．

3（小噺その２）この定義では，共通鍵暗号の暗号文を senc(message, key) としています．この定義を見て，なんだか気持
ち悪いと感じる方も多いのではないでしょうか．暗号界隈では引数を「鍵，平文」の順番で書くことが多いらしいのですが，形式
検証界隈では「平文，鍵」の順で書かれることが多いです．

図 5: コード 1の Aのセッション鍵の機密性についての攻撃トレース．

3.2 公開鍵暗号を使ってみよう
上の例を機密性を満たすように修正していきましょう．通信路に平文でセッション鍵の要素となる情報
（NA）を流していたことが攻撃につながっていたため，公開鍵暗号を用いて通信路に情報を流すように
したいと思います．次のプロトコル P ′を考えてみます．ここで，pkB , skB はBの公開鍵と秘密鍵です．

Generate NA

Alice

cA

Bob

Accept
skA = hash(NA)

Accept
skB = hash(NA)

Compute
cA = aenc(NA,pkB)

Compute
NA = adec(cA,skB)

図 6: 二者間鍵交換プロトコル P ′．

公開鍵暗号は次のコンストラクタとデストラクタで定義されます．
1 (* Asymmetric key encryption *)
2 fun pk(bitstring):bitstring.
3 fun aenc(bitstring, bitstring):bitstring.
4 reduc forall message:bitstring, key:bitstring; adec(aenc(message, pk(key)), key) =

message.

公開鍵 pkを用いたメッセージ mの暗号文は aenc(m, pk)，復号鍵 skを用いた暗号文 cの復号結果は
adec(c, sk)で表されます（正しい復号鍵が使われた時に限り，正しい復号結果を得ることができます）．

演習 3.2. コード 2の ??? 部分を埋めて，プロトコル P ′を検証するコードを完成させてください．
Code 2: 二者間鍵交換プロトコル P ′

1 (* Setting of channels *)
2 free c:channel.
3

4 (* Hash function *)
5 fun hash(bitstring):bitstring.
6

7 (* Symmetric key encryption *)
8 fun senc(bitstring, bitstring):bitstring.
9 reduc forall m:bitstring, k:bitstring; sdec(senc(m, k), k) = m.

10

11 (* Asymmetric key encryption *)
12 fun pk(bitstring):bitstring.
13 fun aenc(bitstring, bitstring):bitstring.
14 reduc forall m:bitstring, k:bitstring; adec(aenc(m, pk(k)), k) = m.
15

16 (* Queries *)
17 free message_for_verification_A:bitstring[private].
18 query attacker(message_for_verification_A).
19 free message_for_verification_B:bitstring[private].
20 query attacker(message_for_verification_B).
21

22 (* Behavior of Alice *)
23 let Alice(pkA:bitstring, pkB:bitstring, skA:bitstring) =
24 new Na:bitstring;

25 let Cipher_A = ???(1) in

26 out(c, Cipher_A);
27 let sk_Alice = hash(Na) in
28 out(c, senc(message_for_verification_A, sk_Alice)).
29

30 (* Behavior of Bob *)
31 let Bob(pkB:bitstring, pkA:bitstring, skB:bitstring) =
32 in(c, Cipher_A_recv:bitstring);

33 let Na = ???(2) in

34 let sk_Bob = hash(Na) in
35 out(c, senc(message_for_verification_B, sk_Bob)).
36 new sskA:bitstring; let spkA = spk(sskA) in out(c, spkA);
37

38 (* Start process *)
39 process new skA:bitstring; let pkA = pk(skA) in out(c, pkA);
40 new skB:bitstring; let pkB = pk(skB) in out(c, pkB);
41 ((!Alice(pkA, pkB, skA)) | !Bob(pkB, pkA, skB))

変数名は次の通りです．
• Aの公開鍵：pkA，Aの秘密鍵：skA（Aのサブプロセスの中でのみ使えます）．
• B の公開鍵：pkB，B の秘密鍵：skB（B のサブプロセスの中でのみ使えます）．

参考解答は FWSのWebページにある code2 complete.pvをご覧ください．以降，参考解答の出力
を想定して話を進めていきます．

code2 complete.pvの出力は次のようになります．
• Query not attacker(message for verification A[]) is true.

• Query not attacker(message for verification B[]) is false.

Aのセッション鍵の機密性は trueになりました．B側のセッション鍵の機密性に対してのみ攻撃が
発見されたようです．攻撃トレースを見てみましょう．

図 7: コード 2の B のセッション鍵の機密性についての攻撃トレース．

トレースによると，攻撃者がBに対して Aになりすましているようです．AからBは公開鍵を用い
て（暗黙的に）認証ができているのですが，Bから Aは何も確かめる術がないため，なりすましが可能
になっています．

3.3 電子署名を使ってみよう＆認証要件を検証してみよう
なりすましを防ぐために，電子署名を用いてみましょう．次のプロトコル P ′′ を考えてみます．

Generate NA

Alice

cA, sA

Bob

Accept
skA = hash(NA)

Accept
skB = hash(NA)

Compute
cA = aenc(NA,pkB)
sA = sign(cA,sskA)

Compute
NA = adec(cA,skB)

Verify
cA = check(sA,spkA)?

図 8: 二者間鍵交換プロトコル P ′′．

spkと sskは署名用の公開鍵（検証鍵）と秘密鍵 (署名鍵)のペアです．電子署名は次のコンストラク
タとデストラクタで定義されます．
1 (* Signatures *)
2 fun spk(bitstring):bitstring.
3 fun sign(bitstring, bitstring):bitstring.
4 reduc forall message:bitstring, key:bitstring; check(sign(message, key), spk(key)) =

message.

署名鍵 sskを用いたメッセージ mへの署名は sign(m, ssk)，署名検証鍵 spkを用いた署名 sの検証結
果は check(s, spk)で表されます（正しい署名の場合はメッセージが返ってきます）．
さらに，攻撃者によるなりすましが可能かどうかを直接検査するために，認証要件を検証してみましょ

う．認証要件はイベントという概念を用いて定義できます．このプロトコルの例では，Bから Aへの認
証要件は「Bがセッション鍵 hash(NA)を受理したならば，それよりも前の段階でAはBとナンスNA

を用いた鍵交換を行おうとしている」という対応関係を見ることにより，検証を実現します．ProVerif
では，
1 event(e1) ==> event(e2)

と書くことで，「イベント e1 が起きているならば，それよりも前にイベント e2 が起きている」という現
象を検証できます．

演習 3.3. コード 3の ??? 部分を埋めて，プロトコル P ′′のセッション鍵の機密性と Bから Aへ
の認証要件を検証するコードを完成させてください．

Code 3: 二者間鍵交換プロトコル P ′′

1 (* Setting of channels *)
2 free c:channel.
3

4 (* Hash function *)
5 fun hash(bitstring):bitstring.
6

7 (* Symmetric key encryption *)
8 fun senc(bitstring, bitstring):bitstring.
9 reduc forall m:bitstring, k: bitstring; sdec(senc(m, k), k) = m.

10

11 (* Asymmetric key encryption *)
12 fun pk(bitstring):bitstring.
13 fun aenc(bitstring, bitstring):bitstring.
14 reduc forall m: bitstring, key:bitstring; adec(aenc(m, pk(key)), key) = m.
15

16 (* Signatures *)
17 fun spk(bitstring):bitstring.
18 fun sign(bitstring, bitstring):bitstring.
19 reduc forall m:bitstring, key:bitstring; check(sign(m, key), spk(key)) = m.
20

21 (* Queries and Events *)
22 free message_for_verification_A:bitstring[private].
23 query attacker(message_for_verification_A).
24 free message_for_verification_B:bitstring[private].
25 query attacker(message_for_verification_B).
26

27 event Accept_B(bitstring).
28 event Send_A(bitstring).
29 query x:bitstring; event(Accept_B(x)) ==> event(Send_A(x)).
30

31 (* Behavior of Alice *)
32 let Alice(pkA:bitstring, pkB:bitstring, spkA:bitstring, spkB:bitstring, skA:

bitstring, sskA:bitstring) =
33 new Na: bitstring;
34 event Send_A(hash(Na));
35 let Cipher_A = aenc(Na, pkB) in

36 let Sig_A = ???(3) in

37 out(c, (Cipher_A, Sig_A));
38 let sk_Alice = hash(Na) in
39 out(c, senc(message_for_verification_A, sk_Alice)).
40

41 (* Behavior of Bob *)
42 let Bob(pkB:bitstring, pkA:bitstring, spkB:bitstring, spkA:bitstring, skB:

bitstring, sskB:bitstring) =
43 in(c, (Cipher_A_recv:bitstring, Sig_A_recv:bitstring));

44 if ???(4) = ???(5) then

45 let Na = adec(Cipher_A_recv, skB) in
46 let sk_Bob = hash(Na) in
47 event Accept(sk_Bob);
48 out(c, senc(message_for_verification_B, sk_Bob)).
49

50 (* Start process *)
51 process new skA:bitstring; let pkA = pk(skA) in out(c, pkA);
52 new skB:bitstring; let pkB = pk(skB) in out(c, pkB);
53 new sskA:bitstring; let spkA = spk(sskA) in out(c, spkA);
54 new sskB:bitstring; let spkB = spk(sskB) in out(c, spkB);
55 ((!Alice(pkA,pkB,spkA,spkB,skA,sskA)) | !Bob(pkB,pkA,spkB,spkA,skB,sskB))

新しく定義された変数名は次の通りです．
• Aの署名検証鍵：spkA，Aの署名鍵：sskA（Aのサブプロセスの中でのみ使えます）．
• B の署名検証鍵：spkB，B の署名鍵：sskB（B のサブプロセスの中でのみ使えます）．

10



こちらも参考解答は code3 complete.pvをご覧ください 4．code3 complete.pvの出力は次のように
なります．

• Query not attacker(message for verification A[]) is true.

• Query not attacker(message for verification B[]) is true.

• Query event(Accept B(x)) ==> event(Send A(x)) is true.

AとBのセッション鍵の機密性およびBからAへの認証性について trueが出力されることが確認でき
ました．

3.4 クエリの種類
補足として，ProVerifの扱える安全性要件について説明します．扱える要件は次の 3つの性質です．

• 到達可能性（reachability）．ある特定のイベントがプロセスの組み合わせによって発生しうるかど
うかを検証します．秘密情報の機密性やプロトコルが最後まで到達可能かどうかなどの要件を検証
できます．

• 対応表明性（correspondence assertions）．あるイベント eが起きたときに，前提となるイベント
e′ が必ず発生しているかどうかを検証します．認証要件などを検証できます．

• 観測等価性（observational equivalence）．2つのプロセスを実行した結果が外部から見て識別可能
かどうかを検証します．プライバシーの要件などを検証できます．

上の例では，機密性の検証を到達可能性のクエリを用いて，認証要件の検証を対応表明性のクエリを用
いて行いました．

4 発展演習
お時間がある方はどうぞ．

演習 4.1. プロトコル P ′を電子署名を用いずにセッション鍵の機密性を満たす（攻撃が発見され
ない）ように修正し，検証してみましょう．ただし，セッション鍵は hash(NA)とします．

演習 4.2. codeDH.pvの???部分を埋めて DH鍵交換プロトコルを検証するコードを作成し，ど
のような攻撃が発見されるか見てみましょう．

Generate x

Alice

gx

Bob

Accept
skA = (gy)x

Generate y

gy

Accept
skB = (gx)y

図 9: DH鍵交換プロトコル．

4（小噺その３）P ′′ もワンパスのシンプルなプロトコルなため，ProVerif のコードは 53 行ほどです．複雑なプロトコルにな
るとどのくらいになるのでしょうか．例えば TLS 1.3 の検証 [KBB17]の github ディレクトリを見てみると，コードの行数（最
大）は 722 行のようです．なお，IETF の TLS1.3 の文書 [Res18] は 160 ページです．

5 おわりに
本ハンズオンでは，簡単な（認証付き）鍵交換プロトコルを例に，ProVerifによる検証方法について紹
介しました．時間の都合上，細かい文法や記述方法については扱いませんでした．まだまだ ProVerifの
能力の 1/10も伝えられていないため，興味を持っていただけた方はマニュアルや既存の検証例などを見
てもらえればと思います．ハンズオンにご参加いただきありがとうございました．

